Rekursio

Seuraava koodi piirtää rekursiivisen kuvion:

```
Sub Piirra(x As Number, y As Number, k As Number)
If k > 1 Then
FillCircle x, y, k
Piirra x+10, y-2*k, k*0.8
End If
End Sub
Piirra 320, 300, 20
```

Koodin tuloksena on seuraava kuva:



Koodissa Piirra on *aliohjelma*, joka piirtää yhden ympyrän ja kutsuu sitten itseään. Muuttujat x, y ja k ovat aliohjelman parametrit. Jokaisen piirtämisen jälkeen x:stä tulee x+10, y:stä tulee y-2*k ja k:sta tulee k*0.8. Piirtäminen päättyy, kun k on 1 tai vähemmän.

Parametrit muuttuvat siis seuraavasti:

x	У	k
320	300	20
330	260	16
340	228	12.8
350	202.4	10.24
360	181.92	8.192

Tehtävä 1

Testaa muuttaa parametreihin x, y ja k liittyviä kaavoja, jotta saat paremman käsityksen rekursion toiminnasta. Näin saat aikaan monenlaisia kuvia. Muista, että k:n pitää pienentyä, jotta kuvion piirtäminen päättyy jossain vaiheessa...

Pöytäliina

Seuraava koodi piirtää pöytäliinan:

```
Sub Piirra(x As Number, y As Number, k As Number)
    If k > 1 Then
        FillCircle x, y, k
        Piirra x-2*k, y-2*k, k*0.5
        Piirra x+2*k, y-2*k, k*0.5
        Piirra x-2*k, y+2*k, k*0.5
        Piirra x+2*k, y+2*k, k*0.5
        End If
End Sub
Piirra 320, 240, 50
```

Koodin tulos on tässä:



Erona aiempaan koodiin tämä koodi *haarautuu* jokaisen ympyrän piirtämisen jälkeen neljään eri suuntaan: ylävasemmalle, yläoikealle, alavasemmalle sekä alaoikealle.

Tehtävä 2

Suunnittele uusia pöytäliinoja muokkaamalla koodia.

Kokeile myös muuttaa, moneenko osaan rekursio haarautuu. Niin voi tulla uusia hienoja kuvioita, ei tosin välttämättä enää pöytäliinoja.

Animaatio

Animaation ohjelmointi onnistuu toistamalla seuraavaa silmukassa:

- 1. tyhjennä näyttö (ClearScreen)
- 2. piirrä kuva uudessa vaiheessa
- 3. päivitä näyttö (DrawScreen)

Esimerkiksi seuraava koodi tuottaa ympyrän, joka kulkee ruudun poikki:

```
For i = 1 To 640
    ClearScreen
    FillCircle i, 240, 50
    DrawScreen
Next i
```

Komento DrawScreen silmukassa aiheuttaa, että kuva piirretään 60 kertaa sekunnissa. Näin syntyy hyvältä näyttävä animaatio.

Jos ympyrä liikkuu mielestäsi liian hitaasti, voit nopeuttaa sitä näin:

```
For i = 1 To 640 Step 5
    ClearScreen
    FillCircle i, 240, 50
    DrawScreen
Next i
```

Tehtävä 3

Tee animaatioita yllä olevan koodien tyyliin.

Voit esimerkiksi tehdä animaation, jossa kaksi ympyrää liikkuu eri suuntiin ja törmää lopuksi. Tai sitten voit tehdä animaation, jossa ympyrät liikkuvat eri nopeuksilla.

Animoitu liina

Seuraava koodi tuottaa animoidun pöytäliinan:

```
Sub Piirra(x As Number, y As Number, k As Number)
    If k > 1 Then
        FillCircle x, y, k
        Piirra x-2*k, y-2*k, k*0.5
        Piirra x+2*k, y-2*k, k*0.5
        Piirra x-2*k, y+2*k, k*0.5
        Piirra x+2*k, y+2*k, k*0.5
    End If
End Sub
Dim a = 10
Do
    ClearScreen
    Piirra 320, 240, a
    a = a + 0.1
    DrawScreen
Loop
```

Animaation aikana pöytäliina liikkuu lähemmäs ja siihen tulee koko ajan uusia yksityiskohtia.

Tehtävä 4

Muokkaa ja tutki tätä koodia. Miten pystyt säätämään yksityiskohtien määrää?

PS. Yksi hyvä muokkaus on liittää satunnaisuutta. Kokeile esim. korvata ympyrän piirto tällä:

FillCircle x+Rnd(-1,1), y+Rnd(-1,1)